Approximation with step functions
The Haar wavelet transform (one level)

Dyadic points

In[1]:=
```
x[j_, k_] := k / 2^j
```

Endpoints of the dyadic intervals at level *j*

In[2]:=
```
X[j_] := Table[x[j, k] , {k, 0, 2^j}]
```

In[3]:=
```
X[3]
```

Out[3]= $\{0, \frac{1}{8}, \frac{1}{4}, \frac{3}{8}, \frac{1}{2}, \frac{5}{8}, \frac{3}{4}, \frac{7}{8}, 1\}$

Midpoints of the dyadic intervals at level *j*

In[4]:=
```
Y[j_] := Table[x[j + 1, 2 k + 1], {k, 0, 2^j - 1}]
```

In[5]:=
```
Y[3]
```

Out[5]= $\{\frac{1}{16}, \frac{3}{16}, \frac{5}{16}, \frac{7}{16}, \frac{9}{16}, \frac{11}{16}, \frac{13}{16}, \frac{15}{16}\}$

the Haar scaling function $\phi(t)$

In[6]:=
```
ϕ[t_] := UnitBox[t - 1 / 2]
```

In[7]:=
```
Plot[ϕ[t], {t, -1, 2}, Filling → Axis]
```

Out[7]=



Translation and dilation of the Haar scaling function

In[9]:=
```
ϕ[j_, k_, t_] := 2^(j/2) ϕ[2^j t - k]
```

In[10]:=
```
Φ[j_, t_] := Table[ϕ[j, k, t], {k, 0, 2^j - 1}]
```

In[11]:=
```
Φ[2, t] /. UnitBox → "ϕ" // MatrixForm
```

Out[11]//MatrixForm=

$$\begin{pmatrix} 2\,\phi\left[\frac{1}{2} - 4\,t\right] \\ 2\,\phi\left[\frac{3}{2} - 4\,t\right] \\ 2\,\phi\left[\frac{5}{2} - 4\,t\right] \\ 2\,\phi\left[\frac{7}{2} - 4\,t\right] \end{pmatrix}$$

In[12]:=
```
Manipulate[
 Plot[Evaluate[ϕ[2, k, t]], {t, -1, 2}, Filling → Axis], {k, -2, 5, 1}]
```

Out[12]=



In[13]:=
```
Plot[{
  ϕ[1, 1, t], ϕ[2, -3, t], ϕ[3, 10, t], ϕ[-1, 0, t]}, {t, -2, 3},
 Filling → Axis,
 PlotStyle → {Black, Red, Green, Blue},
 PlotLegends → "Expressions"]
```

Out[13]=



the Haar approximation coefficients $a_{j,k} = \langle f \mid \phi_{j,k} \rangle$

In[14]:= $a_\phi[f\_, j\_, k\_] := \int_{x[j,k]}^{x[j,k+1]} f(t)\, \phi(j, k, t)\, dt$

In[15]:=
```
naϕ[f_, j_, k_] := NIntegrate[f[t] ϕ[j, k, t], {t, x[j, k], x[j, k + 1]}]
```

In[16]:= $a_\phi[f\_, j\_] := \text{Table}[a_\phi[f, j, k], \{k, 0, 2^j - 1\}]$

In[17]:= $na_\phi[f\_, j\_] := \text{Table}[na_\phi[f, j, k], \{k, 0, 2^j - 1\}]$

an example for approximation

In[18]:=
```
f[t_] := t Sin[10 t]
```

In[19]:= $a_\phi[f, 2]$

Out[19]= $\left\{\frac{1}{100}\left(-5\,\text{Cos}\left[\frac{5}{2}\right] + 2\,\text{Sin}\left[\frac{5}{2}\right]\right),\ \frac{1}{100}\left(5\,\text{Cos}\left[\frac{5}{2}\right] - 10\,\text{Cos}[5] - 2\,\text{Sin}\left[\frac{5}{2}\right] + 2\,\text{Sin}[5]\right),\ \frac{1}{100}\left(10\,\text{Cos}[5] - 15\,\text{Cos}\left[\frac{15}{2}\right] - 2\,\text{Sin}[5] + 2\,\text{Sin}\left[\frac{15}{2}\right]\right),\ \frac{1}{100}\left(15\,\text{Cos}\left[\frac{15}{2}\right] - 20\,\text{Cos}[10] - 2\,\text{Sin}\left[\frac{15}{2}\right] + 2\,\text{Sin}[10]\right)\right\}$

In[20]:= $na_\phi[f, 2]$

Out[20]= {0.0520266, -0.0995713, 0.0143094, 0.190169}

Approximation of *f* on level *j* (= projection into the vector space V_j)

In[21]:= $\text{approx}_\phi[f\_, j\_, t\_] := \text{Total}[a_\phi[f, j]\, \Phi[j, t]]$

In[22]:= $\text{napprox}_\phi[f\_, j\_, t\_] := \text{Total}[na_\phi[f, j]\, \Phi[j, t]]$

In[23]:= $\text{approx}_\phi[f, 3, t]$ /. UnitBox → "ϕ"

Out[23]=
$$-\frac{1}{50}\phi\left[\frac{1}{2} - 8t\right]\left(5\,\text{Cos}\left[\frac{5}{4}\right] - 4\,\text{Sin}\left[\frac{5}{4}\right]\right) +$$
$$\frac{1}{50}\phi\left[\frac{3}{2} - 8t\right]\left(5\,\text{Cos}\left[\frac{5}{4}\right] - 10\,\text{Cos}\left[\frac{5}{2}\right] - 4\,\text{Sin}\left[\frac{5}{4}\right] + 4\,\text{Sin}\left[\frac{5}{2}\right]\right) +$$
$$\frac{1}{50}\phi\left[\frac{5}{2} - 8t\right]\left(10\,\text{Cos}\left[\frac{5}{2}\right] - 15\,\text{Cos}\left[\frac{15}{4}\right] - 4\,\text{Sin}\left[\frac{5}{2}\right] + 4\,\text{Sin}\left[\frac{15}{4}\right]\right) +$$
$$\frac{1}{50}\phi\left[\frac{7}{2} - 8t\right]\left(15\,\text{Cos}\left[\frac{15}{4}\right] - 20\,\text{Cos}[5] - 4\,\text{Sin}\left[\frac{15}{4}\right] + 4\,\text{Sin}[5]\right) +$$
$$\frac{1}{50}\phi\left[\frac{9}{2} - 8t\right]\left(20\,\text{Cos}[5] - 25\,\text{Cos}\left[\frac{25}{4}\right] - 4\,\text{Sin}[5] + 4\,\text{Sin}\left[\frac{25}{4}\right]\right) +$$
$$\frac{1}{50}\phi\left[\frac{11}{2} - 8t\right]\left(25\,\text{Cos}\left[\frac{25}{4}\right] - 30\,\text{Cos}\left[\frac{15}{2}\right] - 4\,\text{Sin}\left[\frac{25}{4}\right] + 4\,\text{Sin}\left[\frac{15}{2}\right]\right) +$$
$$\frac{1}{50}\phi\left[\frac{13}{2} - 8t\right]\left(30\,\text{Cos}\left[\frac{15}{2}\right] - 35\,\text{Cos}\left[\frac{35}{4}\right] - 4\,\text{Sin}\left[\frac{15}{2}\right] + 4\,\text{Sin}\left[\frac{35}{4}\right]\right) +$$
$$\frac{1}{50}\phi\left[\frac{15}{2} - 8t\right]\left(35\,\text{Cos}\left[\frac{35}{4}\right] - 40\,\text{Cos}[10] - 4\,\text{Sin}\left[\frac{35}{4}\right] + 4\,\text{Sin}[10]\right)$$

In[24]:=
```
napprox_ϕ[f, 3, t] /. UnitBox → "ϕ"
```

Out[24]=

$$0.0443865\,\phi\left[\frac{1}{2}-8\,t\right]+0.16372\,\phi\left[\frac{3}{2}-8\,t\right]-0.00766359\,\phi\left[\frac{5}{2}-8\,t\right]-0.390622\,\phi\left[\frac{7}{2}-8\,t\right]-0.3122\,\phi\left[\frac{9}{2}-8\,t\right]+0.369438\,\phi\left[\frac{11}{2}-8\,t\right]+0.729511\,\phi\left[\frac{13}{2}-8\,t\right]+0.0311656\,\phi\left[\frac{15}{2}-8\,t\right]$$

In[25]:=
```
Plot[Evaluate[{f[t], napprox_ϕ[f, 4, t]}], {t, 0, 1},
 Filling → Axis , PlotStyle → {Black, Blue}]
```

Out[25]=



comparing two neighboring levels of resolution

In[26]:=
```
Plot[Evaluate[{napprox_ϕ[f, 3, t], napprox_ϕ[f, 4, t]}], {t, 0, 1},
 Filling → Axis, PlotStyle → {Red, Blue}]
```

Out[26]=



In[27]:=
```
a[n_] := na_ϕ[f, n]
```

In[28]:=
```
aa[n_] := Transpose[{Y[n], a[n]}]
```

In[29]:=
```
ListPlot[aa[3], Filling → Axis]
```

Out[29]=



In[31]:=
```
Manipulate[
 ListPlot[aa[n], Filling → Axis], {n, 1, 7, 1}]
```

Out[31]=



the Haar wavelet function $\psi(t)$

In[32]:=
```
ψ[t_] := UnitBox[2 t - 1 / 2] - UnitBox[2 t - 3 / 2]
```

In[33]:=
```
Plot[ψ[t], {t, -2, 3}, Filling → Axis]
```

Out[33]=



Translation and dilation of the Haar wavelet function

In[34]:=
```
ψ[j_, k_, t_] := 2^(j/2) ψ[2^j t - k]
```

In[35]:=
```
Manipulate[
 Plot[Evaluate[ψ[2, k, t]], {t, -1, 2}, Filling → Axis], {k, -2, 5, 1}]
```

Out[35]=



In[36]:=
```
Plot[{ψ[1, 1, t], ψ[2, -3, t], ψ[3, 10, t], ψ[-1, 0, t]}, {t, -1, 2},
 Filling → Axis,
 PlotStyle → {Black, Red, Green, Blue},
 PlotLegends → "Expressions"]
```

Out[36]=



In[37]:=
```
Ψ[j_, t_] := Table[ψ[j, k, t], {k, 0, 2^j - 1}]
```

the Haar detail coefficients $d_{j,k} = \langle f \mid \psi_{j,k} \rangle$

In[38]:= $d_\psi[f\_, j\_, k\_] := \int_{x[j,k]}^{x[j,k+1]} f(t)\, \psi(j, k, t)\, dt$

In[39]:=
```
nd_ψ[f_, j_, k_] := NIntegrate[f[t] ψ[j, k, t], {t, x[j, k], x[j, k + 1]}]
```

In[40]:=
```
d_ψ[f_, j_] := Table[d_ψ[f, j, k], {k, 0, 2^j - 1}]
```

In[41]:=
```
nd_ψ[f_, j_] := Table[nd_ψ[f, j, k], {k, 0, 2^j - 1}]
```

In[42]:=
```
d[n_] := nd_ψ[f, n]
```

In[43]:=
```
dd[n_] := Transpose[{Y[n], d[n]}]
```

In[44]:=
```
ListPlot[dd[3], Filling → Axis, DataRange → {0, 3}]
```

Out[44]=



In[45]:=
```
Manipulate[
 ListPlot[dd[n], Filling → Axis], {n, 1, 7, 1}]
```

Out[45]=



Detail on level j (= projection into the vector space W_j)

In[47]:=
```
detailψ[f_, j_, t_] := Total[dψ[f, j] Ψ[j, t]]
```

In[48]:=
```
ndetailψ[f_, j_, t_] := Total[ndψ[f, j] Ψ[j, t]]
```

In[49]:=
```
Plot[Evaluate[{f[t], napproxϕ[f, 5, t], ndetailψ[f, 4, t],
   napproxϕ[f, 4, t] + ndetailψ[f, 4, t]}], {t, 0, 1},
 Filling → Axis, PlotStyle → {Black, Red, Green, Blue}]
```

Out[49]=



In[50]:=
```
Plot[Evaluate[{
   napprox_ϕ[f, 4, t] - (napprox_ϕ[f, 3, t] + ndetail_ψ[f, 3, t])}], {t, 0, 1},
 Filling → Axis, PlotStyle → {Red}]
```

Out[50]=



In[51]:=
```
Plot[Evaluate[{napprox_ϕ[f, 5, t], ndetail_ψ[f, 5, t]}], {t, 0, 1},
 Filling → Axis, PlotRange → All, PlotStyle → {Blue, Green}]
```

Out[51]=



the Hadamard matrix

In[52]:=
```
H = {{1, 1}, {1, -1}} / Sqrt[2]; H // MatrixForm
```

Out[52]//MatrixForm=

$$\begin{pmatrix} \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} \end{pmatrix}$$

In[53]:=
```
vec = {a, b}; vec // MatrixForm
```

Out[53]//MatrixForm=

$$\begin{pmatrix} a \\ b \end{pmatrix}$$

In[54]:=
```
H.vec // MatrixForm
```

Out[54]//MatrixForm=

$$\begin{pmatrix} \frac{a}{\sqrt{2}} + \frac{b}{\sqrt{2}} \\ \frac{a}{\sqrt{2}} - \frac{b}{\sqrt{2}} \end{pmatrix}$$

In[55]:=
```
H.H // MatrixForm
```
Out[55]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

The Hadamard matrix *H* is its own inverse.
It is also symmetric, so it is the matrix of an orthogonal transform

In[56]:=
```
vec2matrix[vec_, k_] := Transpose[Partition[vec, k]]
```

In[57]:=
```
vec2matrix[{a, b, c, d, e, f, g, h}, 2] // MatrixForm
```
Out[57]//MatrixForm=

$$\begin{pmatrix} a & c & e & g \\ b & d & f & h \end{pmatrix}$$

In[58]:=
```
vec2matrix[{a, b, c, d, e, f, g, h}, 4] // MatrixForm
```
Out[58]//MatrixForm=

$$\begin{pmatrix} a & e \\ b & f \\ c & g \\ d & h \end{pmatrix}$$

In[59]:=
```
a[3]
```
Out[59]=
```
{0.015693, 0.0578837, -0.00270949,
 -0.138106, -0.110379, 0.130616, 0.257921, 0.0110187}
```

In[60]:=
```
vec2matrix[a[3], 2] // MatrixForm
```
Out[60]//MatrixForm=

$$\begin{pmatrix} 0.015693 & -0.00270949 & -0.110379 & 0.257921 \\ 0.0578837 & -0.138106 & 0.130616 & 0.0110187 \end{pmatrix}$$

In[61]:=
```
H.vec2matrix[a[3], 2] // MatrixForm
```
Out[61]//MatrixForm=

$$\begin{pmatrix} 0.0520266 & -0.0995713 & 0.0143094 & 0.190169 \\ -0.0298334 & 0.0957395 & -0.17041 & 0.174586 \end{pmatrix}$$

In[62]:=
```
a[2]
```
Out[62]=
```
{0.0520266, -0.0995713, 0.0143094, 0.190169}
```

In[63]:=
```
d[2]
```
Out[63]=
```
{-0.0298334, 0.0957395, -0.17041, 0.174586}
```

the Haar transform (one level)

In[64]:=
```
htrans[vec_] := Flatten[H.vec2matrix[vec, 2]]
```

In[65]:=
```
htrans[{a, b, c, d}]
```

Out[65]= $\left\{\frac{a}{\sqrt{2}}+\frac{b}{\sqrt{2}}, \frac{c}{\sqrt{2}}+\frac{d}{\sqrt{2}}, \frac{a}{\sqrt{2}}-\frac{b}{\sqrt{2}}, \frac{c}{\sqrt{2}}-\frac{d}{\sqrt{2}}\right\}$

In[66]:=
```
hta[n_] := htrans[a[n]]
```

In[67]:=
```
hta[5]
```

Out[67]=
```
{0.00312981, 0.0190635, 0.0384352, 0.0434248, 0.0215932, -0.025425, -0.0800371,
 -0.115274, -0.107376, -0.0487239, 0.0452736, 0.139445, 0.192017, 0.172738,
 0.0782408, -0.062658, -0.00232393, -0.00522197, -0.00386149, 0.00179256,
 0.00908064, 0.0138514, 0.0125589, 0.00426221, -0.00847883, -0.0203529,
 -0.0254922, -0.0202123, -0.00506341, 0.0148312, 0.0314833, 0.0372368}
```

In[68]:=
```
Manipulate[
 ListPlot[hta[n], Filling → Axis, PlotRange → All], {n, 1, 7, 1}]
```

Out[68]=



In[69]:=
```
Plot[{
   Evaluate[
    Total[hta[6] [[1 ;; 32]] Φ[5, t]]],
   Evaluate[
    Total[hta[6] [[33 ;; 64]] Ψ[5, t]]]}
  , {t, 0, 1},
 Filling → Axis, PlotRange → All]
```

Out[69]=



In[70]:=
```
Plot[Evaluate[
   Total[hta[6] Join[Φ[5, t], Ψ[5, t]]]], {t, 0, 1},
 Filling → Axis, PlotRange → All]
```

Out[70]=



In[71]:=
```
Plot[Evaluate[Total[a[6] Φ[6, t]]], {t, 0, 1},
 Filling → Axis, PlotRange → All]
```

Out[71]=



Haar scaling and wavelet equations

In[72]:=
```
H.{{ϕ[j + 1, 2 k, t]}, {ϕ[j + 1, 2 k + 1, t]}} // MatrixForm
```

Out[72]//MatrixForm=

$$\begin{pmatrix} 2^{-\frac{1}{2}+\frac{1+j}{2}} \text{UnitBox}\left[\frac{1}{2} + 2k - 2^{1+j} t\right] + 2^{-\frac{1}{2}+\frac{1+j}{2}} \text{UnitBox}\left[\frac{3}{2} + 2k - 2^{1+j} t\right] \\ 2^{-\frac{1}{2}+\frac{1+j}{2}} \text{UnitBox}\left[\frac{1}{2} + 2k - 2^{1+j} t\right] - 2^{-\frac{1}{2}+\frac{1+j}{2}} \text{UnitBox}\left[\frac{3}{2} + 2k - 2^{1+j} t\right] \end{pmatrix}$$

In[73]:=
```
{{ϕ[j, k, t]}, {ψ[j, k, t]}} // MatrixForm
```

Out[73]//MatrixForm=

$$\begin{pmatrix} 2^{j/2} \text{UnitBox}\left[\frac{1}{2} + k - 2^{j} t\right] \\ 2^{j/2} \left(\text{UnitBox}\left[\frac{1}{2} - 2\left(-k + 2^{j} t\right)\right] - \text{UnitBox}\left[\frac{3}{2} - 2\left(-k + 2^{j} t\right)\right]\right) \end{pmatrix}$$

In[74]:=
```
Simplify[%% - %] // MatrixForm
```

Out[74]//MatrixForm=

$$\begin{pmatrix} 0 \\ 0 \end{pmatrix}$$

Haar analyis

Haar scaling equation	$\phi(j, k, t) = \frac{\phi(j+1, 2k, t) + \phi(j+1, 2k+1, t)}{\sqrt{2}}$
Haar wavelet equation	$\psi(j, k, t) = \frac{\phi(j+1, 2k, t) - \phi(j+1, 2k+1, t)}{\sqrt{2}}$
approximation coefficients	$a(f, j, k) = \frac{a(f, j+1, 2k) + a(f, j+1, 2k+1)}{\sqrt{2}}$
wavelet coefficients	$d(f, j, k) = \frac{a(f, j+1, 2k) - a(f, j+1, 2k+1)}{\sqrt{2}}$

the inverse Haar transform

In[75]:=
```
invhtrans[vec_] := Flatten[vec2matrix[vec, Length[vec] / 2].H]
```

In[76]:=
```
invhtrans[htrans[{a, b, c, d}]]
```

Out[76]=
$$\left\{\frac{\frac{a}{\sqrt{2}}-\frac{b}{\sqrt{2}}}{\sqrt{2}}+\frac{\frac{a}{\sqrt{2}}+\frac{b}{\sqrt{2}}}{\sqrt{2}},\ -\frac{\frac{a}{\sqrt{2}}-\frac{b}{\sqrt{2}}}{\sqrt{2}}+\frac{\frac{a}{\sqrt{2}}+\frac{b}{\sqrt{2}}}{\sqrt{2}},\ \frac{\frac{c}{\sqrt{2}}-\frac{d}{\sqrt{2}}}{\sqrt{2}}+\frac{\frac{c}{\sqrt{2}}+\frac{d}{\sqrt{2}}}{\sqrt{2}},\ -\frac{\frac{c}{\sqrt{2}}-\frac{d}{\sqrt{2}}}{\sqrt{2}}+\frac{\frac{c}{\sqrt{2}}+\frac{d}{\sqrt{2}}}{\sqrt{2}}\right\}$$

In[77]:=
```
Simplify[%]
```

Out[77]= {a, b, c, d}

Haar synthesis

$$\phi(j+1, 2k, t) = \frac{\phi(j,k,t) + \psi(j,k,t)}{\sqrt{2}}$$

$$\phi(j+1, 2k+1, t) = \frac{\phi(j,k,t) - \psi(j,k,t)}{\sqrt{2}}$$

$$a(f, j+1, 2k) = \frac{a(f,j,k) + d(f,j,k)}{\sqrt{2}}$$

$$a(f, j+1, 2k+1) = \frac{a(f,j,k) - d(f,j,k)}{\sqrt{2}}$$

an example

In[78]:= a[5]

Out[78]= {0.000569845, 0.00385638, 0.00978741, 0.0171724, 0.0244473, 0.0299083, 0.0319735, 0.0294384, 0.0216897, 0.00884773, -0.00818379, -0.0277726, -0.0477142, -0.0654753, -0.078497, -0.0845247, -0.0819219, -0.069931, -0.0488447, -0.0200614, 0.0139876, 0.050039, 0.0843105, 0.112895, 0.132196, 0.139357, 0.132632, 0.111657, 0.0775867, 0.0330626, -0.0179756, -0.0706363}

In[79]:= hta[5]

Out[79]= {0.00312981, 0.0190635, 0.0384352, 0.0434248, 0.0215932, -0.025425, -0.0800371, -0.115274, -0.107376, -0.0487239, 0.0452736, 0.139445, 0.192017, 0.172738, 0.0782408, -0.062658, -0.00232393, -0.00522197, -0.00386149, 0.00179256, 0.00908064, 0.0138514, 0.0125589, 0.00426221, -0.00847883, -0.0203529, -0.0254922, -0.0202123, -0.00506341, 0.0148312, 0.0314833, 0.0372368}

In[80]:= ListPlot[hta[5], Filling → Axis]

Out[80]=


In[81]:= ihta[n_] := invhtrans[hta[n]]

In[82]:= ListPlot[ihta[5], Filling → Axis]

Out[82]=


In[83]:= **a[5] - ihta[5]**

Out[83]= $\{1.0842\times10^{-19}, 4.33681\times10^{-19}, 1.73472\times10^{-18}, 3.46945\times10^{-18}, 3.46945\times10^{-18}, 6.93889\times10^{-18}, 0., 6.93889\times10^{-18}, 3.46945\times10^{-18}, 0., 0., -6.93889\times10^{-18}, 0., -1.38778\times10^{-17}, -1.38778\times10^{-17}, -2.77556\times10^{-17}, -1.38778\times10^{-17}, -1.38778\times10^{-17}, -6.93889\times10^{-18}, -3.46945\times10^{-18}, 3.46945\times10^{-18}, 6.93889\times10^{-18}, 1.38778\times10^{-17}, 1.38778\times10^{-17}, 2.77556\times10^{-17}, 0., 2.77556\times10^{-17}, 1.38778\times10^{-17}, 1.38778\times10^{-17}, -6.93889\times10^{-18}, -6.93889\times10^{-18}, -1.38778\times10^{-17}\}$

In[84]:= **Chop[%]**

Out[84]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}